Jackson Electrostatics - 4 09-17-17

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[26]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

This is the last of four notebooks dealing with electrostatic topics and calculations as treated in Jackson *Classical Electrodynamics.*

This notebook focuses on the solution of Poisson’s equation for Green’s function in spherical coordinates. A general form for the Green’s function satisfying Dirichlet boundary conditions is obtained in the region between two concentric spherical conductors. Some special cases are obtained by calculating limits of the general case.

Applications of these results are made to three problems

- Conducting sphere with external point charge
- Conducting sphere with external ring of charge
- Conducting sphere with internal line charge

0 Green’s function in spherical coordinates

The Green’s function relevant to spherically symmetric electrostatic potential problems satisfies the partial differential equations

$$\nabla_{\mathbb{r}}^2 G(\mathbb{r},\ \hat{\mathbb{r}}) = -4\pi\,\delta(\mathbb{r}-\hat{\mathbb{r}}) = \frac{1}{r^2}\,\delta(r-\hat{r})\,\delta\left(\cos(\theta)-\cos(\hat{\theta})\right)\delta\left(\phi-\hat{\phi}\right) \tag{1}$$

Under the ansatz of separable variables, Jackson shows in Chapter 3 that the Green’s function can be represented by the expansion

$$G(\mathbb{r},\ \hat{\mathbb{r}}) = \sum_{l=0}^{\infty} \sum_{m=0}^{\infty} A_{l,m}\left(\hat{\theta},\ \hat{\phi}\right) g_l(\mathbb{r},\ \hat{\mathbb{r}})\, Y_{l,m}(\theta,\ \phi) \tag{2}$$

where the $Y_{l,m}(\theta,\ \phi)$ are spherical harmonics and $g_l(\mathbb{r},\hat{\mathbb{r}})$ satisfies

$$\frac{1}{r} \frac{d^2}{dr^2}(r\, g_l(r,\ \hat{r})) - \frac{l(l+1)}{r^2} g_l(r,\ \hat{r}) = -\frac{4\pi}{r^2}\delta(r - \hat{r}) \tag{3}$$

In this notebook, I will work through the details of solving equation (3) and then use

$$\Phi(\mathbb{r}) = \int_V d^3\hat{r}\, \rho(\hat{\mathbb{r}})\, G(\mathbb{r},\ \hat{\mathbb{r}})$$
$$= \sum_{l=0}^{\infty} \sum_{m=-l}^{l} \int_0^{2\pi} d\hat{\phi} \int_0^{\pi} \sin(\hat{\theta})\, d\hat{\theta} \int_0^{\infty} d\hat{r}\, \hat{r}^2\, \rho(\hat{\mathbb{r}})\, g_l(\mathbb{r},\ \hat{\mathbb{r}})\, Y_{l,m}(\theta,\ \phi) \tag{4}$$

to solve some potential problem involving spherically symmetric conductors in the presence of charge distributions.

1 Solving for the radial component of the Green's function

I make use of Notation package so as to obtain intermediate expressions that resemble those that would obtain from a hand calculation that utilized standard physics notation.

In[3]:=
```
<< Notation`
```

In[4]:=
```
Symbolize[ 𝕣̂ ];
Symbolize[ r̂ ];
Symbolize[ θ̂ ];
Symbolize[ ϕ̂ ];
Symbolize[ r> ];
Symbolize[ r< ];
Symbolize[ g> ];
Symbolize[ g< ];
Symbolize[ A< ];
Symbolize[ B< ];
Symbolize[ A> ];
Symbolize[ B> ];
Symbolize[ G< ];
Symbolize[ G> ];
```

In[13]:=
```
Notation[ H[a_] ⟺ HeavisideTheta[a_] ];
Notation[ δ[a_] ⟺ DiracDelta[a_] ];
Notation[ δ'[a_] ⟹ DiracDelta'[a_] ]
```

I start with equation (3)

```
w1[1] = 1/r D[r g[r, r̂], {r, 2}] - ℓ (ℓ + 1)/r^2 g[r, r̂] == - 4 π/r^2 δ[r - r̂] // Expand
```

$$-\frac{\ell\, g[r,\hat{r}]}{r^2}-\frac{\ell^2\, g[r,\hat{r}]}{r^2}+\frac{2\, g^{(1,0)}[r,\hat{r}]}{r}+g^{(2,0)}[r,\hat{r}]==-\frac{4\pi\,\delta[r-\hat{r}]}{r^2}$$

The homogeneous equation can be solved immediately

```
w1[2] = DSolve[w1[1][[1]] == 0, g[r, r̂], r, Assumptions → ℓ ≥ 0][[1, 1]] // RE
```

$$g[r,\hat{r}]==r^{\frac{1}{2}\,i\,\sqrt{\ell}\,\sqrt{1+\ell}\left(\frac{i}{\sqrt{\ell}\,\sqrt{1+\ell}}-i\sqrt{4+\frac{1}{\ell(1+\ell)}}\right)}\,C[1]+r^{\frac{1}{2}\,i\,\sqrt{\ell}\,\sqrt{1+\ell}\left(\frac{i}{\sqrt{\ell}\,\sqrt{1+\ell}}+i\sqrt{4+\frac{1}{\ell(1+\ell)}}\right)}\,C[2]$$

This can be simplified

```
w1[2] = (w1[2] // ExpandAll) /. {C[1] → A, C[2] → B}
```

$$g[r,\hat{r}]==B\,r^{-\frac{1}{2}-\frac{1}{2}\sqrt{\ell}\,\sqrt{1+\ell}\,\sqrt{4+\frac{1}{\ell+\ell^2}}}+A\,r^{-\frac{1}{2}+\frac{1}{2}\sqrt{\ell}\,\sqrt{1+\ell}\,\sqrt{4+\frac{1}{\ell+\ell^2}}}$$

```
w1[2] = MapEqn[Simplify[#, ℓ ≥ 0] &, w1[2]]
```

$$g[r,\hat{r}]==B\,r^{-1-\ell}+A\,r^{\ell}$$

The constants of integration differ depending on whether r < $\hat{r}$ or not. There are four constants of integration — two determined by boundary conditions as r → 0, r → ∞ and two determined by matching conditions at r = $\hat{r}$.

To derive the matching conditions I write

```
w1[3] = g[r, r̂] == g<[r, r̂] H[r̂ - r] + g>[r, r̂] H[r - r̂]
```

$$g[r,\hat{r}]==g_{>}[r,\hat{r}]\,H[r-\hat{r}]+g_{<}[r,\hat{r}]\,H[-r+\hat{r}]$$

where $g_{>}[r, \hat{r}]$ pertains to r > $\hat{r}$ and $g_{<}[r, \hat{r}]$ pertains to r < $\hat{r}$.

I have to be careful when substituting this ansatz into w1[1] so that the relationships between derivatives of δ and H are imposed correctly. The steps are

```
dg[1] = D[g[r, r̂], r] == (D[g[r, r̂], r] /.
     g → Function[{r, r̂}, g<[r, r̂] H[r̂ - r] + g>[r, r̂] H[r - r̂]] // Expand)
```

$$g^{(1,0)}[r,\hat{r}] == \delta[r-\hat{r}]\, g_{>}[r,\hat{r}] - \delta[-r+\hat{r}]\, g_{<}[r,\hat{r}] + H[r-\hat{r}]\, g_{>}^{(1,0)}[r,\hat{r}] + H[-r+\hat{r}]\, g_{<}^{(1,0)}[r,\hat{r}]$$

Impose the symmetry condition for the δ-function

```
dg[2] = dg[1] /. δ[-r + r̂] → δ[r - r̂] /. a_ δ[r - r̂] :> (a /. r → r̂) δ[r - r̂]
```

$$g^{(1,0)}[r,\hat{r}] == \delta[r-\hat{r}]\, g_{>}[\hat{r},\hat{r}] - \delta[r-\hat{r}]\, g_{<}[\hat{r},\hat{r}] + H[r-\hat{r}]\, g_{>}^{(1,0)}[r,\hat{r}] + H[-r+\hat{r}]\, g_{<}^{(1,0)}[r,\hat{r}]$$

Calculate the second derivative

```
d2g[1] = MapEqn[D[#, r] &, dg[2]]
```

$$g^{(2,0)}[r,\hat{r}] == g_{>}[\hat{r},\hat{r}]\, \mathrm{DiracDelta}'[r-\hat{r}] - g_{<}[\hat{r},\hat{r}]\, \mathrm{DiracDelta}'[r-\hat{r}] + \delta[r-\hat{r}]\, g_{>}^{(1,0)}[r,\hat{r}] - \delta[-r+\hat{r}]\, g_{<}^{(1,0)}[r,\hat{r}] + H[r-\hat{r}]\, g_{>}^{(2,0)}[r,\hat{r}] + H[-r+\hat{r}]\, g_{<}^{(2,0)}[r,\hat{r}]$$

```
d2g[2] = d2g[1] /. δ[-r + r̂] → δ[r - r̂] /. a_ δ[r - r̂] :> (a /. r → r̂) δ[r - r̂]
```

$$g^{(2,0)}[r,\hat{r}] == g_{>}[\hat{r},\hat{r}]\, \mathrm{DiracDelta}'[r-\hat{r}] - g_{<}[\hat{r},\hat{r}]\, \mathrm{DiracDelta}'[r-\hat{r}] + \delta[r-\hat{r}]\, g_{>}^{(1,0)}[\hat{r},\hat{r}] - \delta[r-\hat{r}]\, g_{<}^{(1,0)}[\hat{r},\hat{r}] + H[r-\hat{r}]\, g_{>}^{(2,0)}[r,\hat{r}] + H[-r+\hat{r}]\, g_{<}^{(2,0)}[r,\hat{r}]$$

```
w1[4] = w1[1] /. (dg[2] // ER) /. (d2g[2] // ER) /.
   g → Function[{r, r̂}, g<[r, r̂] H[r̂ - r] + g>[r, r̂] H[r - r̂]]
```

$$-\frac{1}{r^2}\ell\left(g_{>}[r,\hat{r}]\, H[r-\hat{r}] + g_{<}[r,\hat{r}]\, H[-r+\hat{r}]\right) - \frac{1}{r^2}\ell^2\left(g_{>}[r,\hat{r}]\, H[r-\hat{r}] + g_{<}[r,\hat{r}]\, H[-r+\hat{r}]\right) + g_{>}[\hat{r},\hat{r}]\, \mathrm{DiracDelta}'[r-\hat{r}] - g_{<}[\hat{r},\hat{r}]\, \mathrm{DiracDelta}'[r-\hat{r}] + \delta[r-\hat{r}]\, g_{>}^{(1,0)}[\hat{r},\hat{r}] + \frac{1}{r} 2\left(\delta[r-\hat{r}]\, g_{>}[\hat{r},\hat{r}] - \delta[r-\hat{r}]\, g_{<}[\hat{r},\hat{r}] + H[r-\hat{r}]\, g_{>}^{(1,0)}[r,\hat{r}] + H[-r+\hat{r}]\, g_{<}^{(1,0)}[r,\hat{r}]\right) - \delta[r-\hat{r}]\, g_{<}^{(1,0)}[\hat{r},\hat{r}] + H[r-\hat{r}]\, g_{>}^{(2,0)}[r,\hat{r}] + H[-r+\hat{r}]\, g_{<}^{(2,0)}[r,\hat{r}] == -\frac{4\pi\,\delta[r-\hat{r}]}{r^2}$$

Match terms on the lhs and rhs. The coefficients of the Heaviside functions reproduce the odes

```
w1[5] =
 {MapEqn[Coefficient[#, H[r - r̂]] &, w1[4]],
  MapEqn[Coefficient[#, H[-r + r̂]] &, w1[4]]}
```

$$\left\{-\frac{\ell\, g_>[r,\hat{r}]}{r^2}-\frac{\ell^2\, g_>[r,\hat{r}]}{r^2}+\frac{2\, g_>^{(1,0)}[r,\hat{r}]}{r}+g_>^{(2,0)}[r,\hat{r}]==0,\right.$$
$$\left.-\frac{\ell\, g_<[r,\hat{r}]}{r^2}-\frac{\ell^2\, g_<[r,\hat{r}]}{r^2}+\frac{2\, g_<^{(1,0)}[r,\hat{r}]}{r}+g_<^{(2,0)}[r,\hat{r}]==0\right\}$$

The coefficients of the δ-functions provide matching conditions —

```
w1[6] = MapEqn[Coefficient[#, DiracDelta'[r - r̂]] &, w1[4]]
```

$$g_>[\hat{r},\hat{r}]-g_<[\hat{r},\hat{r}]==0$$

which implied the radial Green’s function must be continuous across the boundary

The second matching condition is

```
w1[7] = MapEqn[Coefficient[#, δ[r - r̂]] &, w1[4]]
```

$$\frac{2\, g_>[\hat{r},\hat{r}]}{r}-\frac{2\, g_<[\hat{r},\hat{r}]}{r}+g_>^{(1,0)}[\hat{r},\hat{r}]-g_<^{(1,0)}[\hat{r},\hat{r}]==-\frac{4\pi}{r^2}$$

or

```
w1[7] = w1[7] /. Solve[w1[6], g>[r̂, r̂]][[1, 1]] /. r → r̂
```

$$g_>^{(1,0)}[\hat{r},\hat{r}]-g_<^{(1,0)}[\hat{r},\hat{r}]==-\frac{4\pi}{\hat{r}^2}$$

This is a jump condition for the derivative of g(r, $\hat{r}$).

I develop the Green’s function specific to the region between two concentric conducting spheres



The inner and outer solutions are

$$\text{w1}[8] = \{\text{w1}[2] \text{ /. } \{g \to g_<, A \to A_<, B \to B_<\}, \text{w1}[2] \text{ /. } \{g \to g_>, A \to A_>, B \to B_>\}\}$$

$$\left\{g_<\left[r, \hat{r}\right] == B_< r^{-1-\ell} + A_< r^{\ell}, g_>\left[r, \hat{r}\right] == B_> r^{-1-\ell} + A_> r^{\ell}\right\}$$

Impose a Dirichlet boundary condition at the surface of the inner conductor

$$\text{w1}[9] = \text{w1}[8][[1]] \text{ /. } r \to a \text{ /. } g_<\left[a, \hat{r}\right] \to 0$$

$$0 == a^{\ell} A_< + a^{-1-\ell} B_<$$

Then

$$\text{w1}[10] = \text{w1}[8][[1]] \text{ /. Solve}[\text{w1}[9], A_<][[1]]$$

$$g_<\left[r, \hat{r}\right] == B_< r^{-1-\ell} - a^{-1-2\ell} B_< r^{\ell}$$

At the outer boundary

$$\text{w1}[11] = \text{w1}[8][[2]] \text{ /. } r \to b \text{ /. } g_>\left[b, \hat{r}\right] \to 0$$

$$0 == A_> b^{\ell} + b^{-1-\ell} B_>$$

$$\text{w1}[12] = \text{w1}[8][[2]] \text{ /. Solve}[\text{w1}[11], A_>][[1]]$$

$$g_>\left[r, \hat{r}\right] == B_> r^{-1-\ell} - b^{-1-2\ell} B_> r^{\ell}$$

Require continuity at $r = \hat{r}$

```
w1[13] = {w1[10] /. r → r̂, w1[12] /. r → r̂}
```

$$\left\{g_{<}[\hat{r}, \hat{r}] == B_{<}\,\hat{r}^{-1-\ell} - a^{-1-2\ell} B_{<}\,\hat{r}^{\ell},\ g_{>}[\hat{r}, \hat{r}] == B_{>}\,\hat{r}^{-1-\ell} - b^{-1-2\ell} B_{>}\,\hat{r}^{\ell}\right\}$$

or

```
w1[14] = w1[13][[1, 2]] == w1[13][[2, 2]]
```

$$B_{<}\,\hat{r}^{-1-\ell} - a^{-1-2\ell} B_{<}\,\hat{r}^{\ell} == B_{>}\,\hat{r}^{-1-\ell} - b^{-1-2\ell} B_{>}\,\hat{r}^{\ell}$$

Radial derivatives are needed for the jump condition

```
w1[15] = {MapEqn[D[#, r] &, w1[8][[1]] ], MapEqn[D[#, r] &, w1[8][[2]] ]} /. r → r̂
```

$$\left\{g_{<}^{(1,0)}[\hat{r}, \hat{r}] == B_{<}\,\hat{r}^{-2-\ell}\,(-1-\ell) + A_{<}\,\hat{r}^{-1+\ell}\,\ell,\ g_{>}^{(1,0)}[\hat{r}, \hat{r}] == B_{>}\,\hat{r}^{-2-\ell}\,(-1-\ell) + A_{>}\,\hat{r}^{-1+\ell}\,\ell\right\}$$

```
w1[16] = w1[7] /. (w1[15] // ER) /. Solve[w1[9], A<][[1]] /. Solve[w1[11], A>][[1]]
```

$$B_{>}\,\hat{r}^{-2-\ell}\,(-1-\ell) - B_{<}\,\hat{r}^{-2-\ell}\,(-1-\ell) - b^{-1-2\ell} B_{>}\,\hat{r}^{-1+\ell}\,\ell + a^{-1-2\ell} B_{<}\,\hat{r}^{-1+\ell}\,\ell == -\frac{4\pi}{\hat{r}^{2}}$$

Finally,

```
w1[17] = Solve[{w1[14], w1[16]}, {B<, B>}][[1]]
```

$$\left\{B_{<} \to -\left(\left(4\,a^{1+2\ell}\,\pi\,\hat{r}^{-1-\ell}\left(-b^{1+2\ell} + \hat{r}^{1+2\ell}\right)\right) \Big/ \left(\left(a^{1+2\ell} - b^{1+2\ell}\right)(1+2\ell)\right)\right),\ B_{>} \to \left(4\,b^{1+2\ell}\,\pi\,\hat{r}^{-1-\ell}\left(-a^{1+2\ell} + \hat{r}^{1+2\ell}\right)\right) \Big/ \left(\left(-a^{1+2\ell} + b^{1+2\ell}\right)(1+2\ell)\right)\right\}$$

Using these results

```
w1["inner and outer conductors"] =
 w1[8] /. Solve[w1[9], A<][[1]] /. Solve[w1[11], A>][[1]] /. w1[17]
```

$$\left\{g_{<}[r, \hat{r}] == -\left(\left(4\,a^{1+2\ell}\,\pi\,r^{-1-\ell}\,\hat{r}^{-1-\ell}\left(-b^{1+2\ell} + \hat{r}^{1+2\ell}\right)\right) \Big/ \left(\left(a^{1+2\ell} - b^{1+2\ell}\right)(1+2\ell)\right)\right) + \left(4\,\pi\,r^{\ell}\,\hat{r}^{-1-\ell}\left(-b^{1+2\ell} + \hat{r}^{1+2\ell}\right)\right) \Big/ \left(\left(a^{1+2\ell} - b^{1+2\ell}\right)(1+2\ell)\right),\ g_{>}[r, \hat{r}] == \left(4\,b^{1+2\ell}\,\pi\,r^{-1-\ell}\,\hat{r}^{-1-\ell}\left(-a^{1+2\ell} + \hat{r}^{1+2\ell}\right)\right) \Big/ \left(\left(-a^{1+2\ell} + b^{1+2\ell}\right)(1+2\ell)\right) - \left(4\,\pi\,r^{\ell}\,\hat{r}^{-1-\ell}\left(-a^{1+2\ell} + \hat{r}^{1+2\ell}\right)\right) \Big/ \left(\left(-a^{1+2\ell} + b^{1+2\ell}\right)(1+2\ell)\right)\right\}$$

I perform a visual check for some nominal parameters.

```
Module[{a = 1, b = 3, r̂ = 2, ℓ = 1, lab},
 lab = Stl@StringForm["Test for a = ``, r̂ = ``, b = ``", a, r̂, b];
 Plot[(-((4 a^(1+2ℓ) π r^(-1-ℓ) r̂^(-1-ℓ) (-b^(1+2ℓ) + r̂^(1+2ℓ))) / ((a^(1+2ℓ) - b^(1+2ℓ)) (1 + 2 ℓ))) +
      (4 π r^ℓ r̂^(-1-ℓ) (-b^(1+2ℓ) + r̂^(1+2ℓ))) / ((a^(1+2ℓ) - b^(1+2ℓ)) (1 + 2 ℓ))) HeavisideTheta[r̂ - r] +
    ((4 b^(1+2ℓ) π r^(-1-ℓ) r̂^(-1-ℓ) (-a^(1+2ℓ) + r̂^(1+2ℓ))) / ((-a^(1+2ℓ) + b^(1+2ℓ)) (1 + 2 ℓ)) -
      (4 π r^ℓ r̂^(-1-ℓ) (-a^(1+2ℓ) + r̂^(1+2ℓ))) / ((-a^(1+2ℓ) + b^(1+2ℓ)) (1 + 2 ℓ))) HeavisideTheta[r - r̂],
  {r, a, b}, AxesLabel → {Stl["r"], Stl["g(r, r̂)"]}, PlotLabel → lab]]
```



Useful special cases are

```
w1["no inner conductor"] =
 {w1["inner and outer conductors"] [[1, 1]] ==
   Limit[w1["inner and outer conductors"] [[1, 2]], a → 0, Assumptions → ℓ ≥ 0],
  w1["inner and outer conductors"] [[2, 1]] ==
   Limit[w1["inner and outer conductors"] [[2, 2]], a → 0, Assumptions → ℓ ≥ 0]}
```

$$\left\{g_<[r, \hat{r}] == \frac{4\pi r^{\ell}\, \hat{r}^{-\ell}\left(\frac{1}{\hat{r}} - b^{-1-2\ell}\, \hat{r}^{2\ell}\right)}{1+2\ell},\ g_>[r, \hat{r}] == \frac{4\pi r^{-\ell}\left(\frac{1}{r} - b^{-1-2\ell}\, r^{2\ell}\right) \hat{r}^{\ell}}{1+2\ell}\right\}$$

```
w1["no outer conductor"] =
 {w1["inner and outer conductors"] [[1, 1]] ==
   Limit[w1["inner and outer conductors"] [[1, 2]], b → ∞, Assumptions → ℓ ≥ 0],
  w1["inner and outer conductors"] [[2, 1]] ==
   Limit[w1["inner and outer conductors"] [[2, 2]], b → ∞, Assumptions → ℓ ≥ 0]}
```

$$\left\{g_<[r, \hat{r}] == \frac{1}{1+2\ell} 4\pi r^{-1-\ell}\left(-a^{1+2\ell} + r^{1+2\ell}\right) \hat{r}^{-1-\ell},\ g_>[r, \hat{r}] == \frac{1}{1+2\ell} 4\pi r^{-1-\ell}\, \hat{r}^{-1-\ell}\left(-a^{1+2\ell} + \hat{r}^{1+2\ell}\right)\right\}$$

```
w1["no conductors"] =
 {w1["no outer conductor"] [[1, 1]] ==
   Limit[w1["no outer conductor"] [[1, 2]], a → 0, Assumptions → ℓ ≥ 0],
  w1["no outer conductor"][[2, 1]] ==
   Limit[w1["no outer conductor"][[2, 2]], a → 0, Assumptions → ℓ ≥ 0]}
```

$$\left\{g_{<}[r, \hat{r}] == \frac{4\pi r^{\ell} \hat{r}^{-\ell}}{\hat{r} + 2\hat{r}\ell}, g_{>}[r, \hat{r}] == \frac{4\pi r^{-\ell} \hat{r}^{\ell}}{r + 2r\ell}\right\}$$

2 Conducting sphere with external point charge



This problem has already been solved in notebook *Jackson Electrostatics - 1 09-16-17* using the method of images. Here, the more general Green's function approach is pursued.

A specific calculation of the potential defined by equation (4) involves a double summation, a triple integral, and various symbolic manipulations. Rather than use Mathematica's Integrate and Sum functions, which I would have to "Inactivate" to prevent premature evaluate, I develop my own notation.

The specific definitions of the summation and integration functions are

$$\mathcal{S}[\ell][f] \equiv \sum_{\ell=0}^{\infty} f$$

$$\mathcal{S}[m][f] \equiv \sum_{m=-\ell}^{\ell} f$$

$$\mathcal{I}[\hat{\phi}][f] \equiv \int_0^{2\pi} f\, d\hat{\phi}$$

$$\mathcal{I}[\hat{\theta}][f] \equiv \int_0^{\pi} \sin(\hat{\theta})\, f\, d\hat{\theta}$$

$$\mathcal{I}_<[\hat{r}][f] \equiv \int_0^{r} \hat{r}^2\, f\, d\hat{r}$$

$$\mathcal{I}_>[\hat{r}][f] \equiv \int_r^{\infty} \hat{r}^2\, f\, d\hat{r}$$

For convenience I symbolize the operators

In[16]:=
```
Symbolize[ 𝓘> ];
Symbolize[ 𝓘< ];
```

and also the parameter representing the location of the point charge on the z-axis

In[18]:=
```
Symbolize[ z0 ];
```

I assign some properties to these operators that moves constant factors outside the operator

In[19]:=
```
Clear[𝓘, 𝓢];
(* move constant factors outside the operator *)
𝓘[x_][a_ b_] /; FreeQ[a, x] := a 𝓘[x][b];
𝓘<[x_][a_ b_] /; FreeQ[a, x] := a 𝓘<[x][b];
𝓘>[x_][a_ b_] /; FreeQ[a, x] := a 𝓘>[x][b];
𝓢[x_][a_ b_] /; FreeQ[a, x] := a 𝓢[x][b];
(* distribute sums *)
𝓘[x_][a_ + b_] := 𝓘[x][a] + 𝓘[x][b];
𝓢[x_][a_ + b_] := 𝓢[x][a] + 𝓢[x][b];
```

Without loss of generality, the point charge is chosen to lie on the z-axis, and the problem is axisymmetric since the charge distribution has no explicit ϕ-dependence

$$\text{def}[\rho\text{Axisymmetric}] = \rho\left[\hat{r}, \hat{\theta}, \hat{\phi}\right] \to q\, \frac{\delta\left[\hat{r} - z_0\right] \delta[\hat{\theta}]}{2\, \pi\, \hat{r}^2\, \text{Sin}[\hat{\theta}]}$$

$$\rho\left[\hat{r}, \hat{\theta}, \hat{\phi}\right] \to \frac{q\, \text{Csc}[\hat{\theta}]\, \delta\left[\hat{r} - z_0\right]\, \delta[\hat{\theta}]}{2\, \pi\, \hat{r}^2}$$

where the $2\,\pi$ is necessary to normalize the $\hat{\phi}$ integration.

I can make a quick check that the normalization is correct

```
normalization[1] = Integrate[
  r̂^2 Integrate[Sin[θ̂] Integrate[(ρ[r̂, θ̂, ϕ̂] /. def[ρAxisymmetric]), {ϕ̂, 0, 2π}],
    {θ̂, 0, π}], {r̂, 0, ∞}, Assumptions → z0 > 0]
```

```
q H[0]
```

For this problem, equation (4) for Φ(r, *θ*) is represented by

```
w2[1] =
 S[ℓ][S[m][I[ϕ̂][I[θ̂][Sin[θ̂] I<[r̂][r̂^2 ρ[r̂, θ̂, ϕ̂] G<[ℓ, m, r̂, θ̂, ϕ̂, r, θ, ϕ]]]]]] +
  S[ℓ][S[m][I[ϕ̂][I[θ̂][Sin[θ̂] I>[r̂][r̂^2 ρ[r̂, θ̂, ϕ̂] G>[ℓ, m, r̂, θ̂, ϕ̂, r, θ, ϕ]]]]]]
```

```
S[ℓ][S[m][I[ϕ̂][I[θ̂][Sin[θ̂] I>[r̂][r̂^2 G>[ℓ, m, r̂, θ̂, ϕ̂, r, θ, ϕ] ρ[r̂, θ̂, ϕ̂]]]]]] +
 S[ℓ][S[m][I[ϕ̂][I[θ̂][Sin[θ̂] I<[r̂][r̂^2 G<[ℓ, m, r̂, θ̂, ϕ̂, r, θ, ϕ] ρ[r̂, θ̂, ϕ̂]]]]]]
```

Since there a single point charge outside the conductor, the appropriate radial component of the Green's function is (derived above as w1["no outer conductor"])

```
w1["no outer conductor"] = {g<[r, r̂] == (4 π r^(-1-ℓ) (-a^(1+2ℓ) + r^(1+2ℓ)) r̂^(-1-ℓ)) / (1 + 2ℓ),
  g>[r, r̂] == (4 π r^(-1-ℓ) r̂^(-1-ℓ) (-a^(1+2ℓ) + r̂^(1+2ℓ))) / (1 + 2ℓ)}
```

$$\left\{g_<[r,\hat{r}] == \frac{1}{1+2\ell}4\pi r^{-1-\ell}\left(-a^{1+2\ell}+r^{1+2\ell}\right)\hat{r}^{-1-\ell},\ g_>[r,\hat{r}] == \frac{1}{1+2\ell}4\pi r^{-1-\ell}\hat{r}^{-1-\ell}\left(-a^{1+2\ell}+\hat{r}^{1+2\ell}\right)\right\}$$

Then

```
def[G<] = G<[ℓ, m, r̂, θ̂, ϕ̂, r, θ, ϕ] → g<[r, r̂] Y[ℓ, m][θ, ϕ] Ys[ℓ, m][θ̂, ϕ̂] /.
   {Y[ℓ, m][θ, ϕ] → Sqrt[(2ℓ + 1)/(4π)] Sqrt[(ℓ - m)!/(ℓ + m)!] P[ℓ, m][Cos[θ]] Exp[I m ϕ],
    Ys[ℓ, m][θ̂, ϕ̂] → Sqrt[(2ℓ + 1)/(4π)] Sqrt[(ℓ - m)!/(ℓ + m)!] P[ℓ, m][Cos[θ̂]] Exp[-I m ϕ̂]} /.
  (w1["no outer conductor"] // ER)
```

$$G_<[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi] \to \left(e^{i m\phi - i m\hat{\phi}} r^{-1-\ell}\left(-a^{1+2\ell}+r^{1+2\ell}\right)\hat{r}^{-1-\ell}(-m+\ell)!\,P[\ell,m][\mathrm{Cos}[\theta]]\,P[\ell,m]\left[\mathrm{Cos}[\hat{\theta}]\right]\right)\Big/(m+\ell)!$$

```
def[G>] = G>[ℓ, m, r̂, θ̂, ϕ̂, r, θ, ϕ] → g>[r, r̂] Y[ℓ, m][θ, ϕ] Ys[ℓ, m][θ̂, ϕ̂] /.
    {Y[ℓ, m][θ, ϕ] → Sqrt[(2ℓ + 1)/(4π)] Sqrt[(ℓ - m)!/(ℓ + m)!] P[ℓ, m][Cos[θ]] Exp[I m ϕ],
     Ys[ℓ, m][θ̂, ϕ̂] → Sqrt[(2ℓ + 1)/(4π)] Sqrt[(ℓ - m)!/(ℓ + m)!] P[ℓ, m][Cos[θ̂]] Exp[-I m ϕ̂]} /.
   (w1["no outer conductor"] // ER)
```

$$G_>[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi] \to \left(e^{i m \phi - i m \hat{\phi}}\, r^{-1-\ell}\, \hat{r}^{-1-\ell} \left(-a^{1+2\ell} + \hat{r}^{1+2\ell}\right) (-m+\ell)!\, P[\ell, m][\mathrm{Cos}[\theta]]\, P[\ell, m]\left[\mathrm{Cos}[\hat{\theta}]\right]\right) \Big/ (m+\ell)!$$

where I have re-expressed the spherical harmonics in terms of Associated Legendre functions to facilitate the calculation of the angular integrals.

Using the explicit forms for the charge density and Green’s function

```
w2[2] = w2[1] //. def[ρAxisymmetric] /. def[G<] /. def[G>]
```

$$\frac{1}{2\pi} q\, \mathcal{S}[\ell]\left[r^{-1-\ell}\, \mathcal{I}_>[\hat{r}]\left[\hat{r}^{-1-\ell}\left(-a^{1+2\ell} + \hat{r}^{1+2\ell}\right) \delta\left[\hat{r} - z_0\right]\right]\right.$$
$$\left.\mathcal{S}[m]\left[\left((-m+\ell)!\, P[\ell, m][\mathrm{Cos}[\theta]]\, \mathcal{I}[\hat{\theta}]\left[\delta[\hat{\theta}]\, P[\ell, m]\left[\mathrm{Cos}[\hat{\theta}]\right]\right] \mathcal{I}[\hat{\phi}]\left[e^{i m \phi - i m \hat{\phi}}\right]\right) \Big/ (m+\ell)!\right]\right] +$$
$$\frac{1}{2\pi} q\, \mathcal{S}[\ell]\left[r^{-1-\ell}\left(-a^{1+2\ell} + r^{1+2\ell}\right) \mathcal{I}_<[\hat{r}]\left[\hat{r}^{-1-\ell}\, \delta\left[\hat{r} - z_0\right]\right] \mathcal{S}[m]\left[\right.\right.$$
$$\left.\left.\left((-m+\ell)!\, P[\ell, m][\mathrm{Cos}[\theta]]\, \mathcal{I}[\hat{\theta}]\left[\delta[\hat{\theta}]\, P[\ell, m]\left[\mathrm{Cos}[\hat{\theta}]\right]\right] \mathcal{I}[\hat{\phi}]\left[e^{i m \phi - i m \hat{\phi}}\right]\right) \Big/ (m+\ell)!\right]\right]$$

Perform the $\hat{\phi}$ integration

```
w2[3] = w2[2] //. Op_[ϕ̂][f_] :> (2 π δK[m, 0])
```

$$q\, \mathcal{S}[\ell]\left[r^{-1-\ell}\, \mathcal{I}_>[\hat{r}]\left[\hat{r}^{-1-\ell}\left(-a^{1+2\ell} + \hat{r}^{1+2\ell}\right) \delta\left[\hat{r} - z_0\right]\right]\right.$$
$$\left.\mathcal{S}[m]\left[\left((-m+\ell)!\, \delta K[m, 0]\, P[\ell, m][\mathrm{Cos}[\theta]]\, \mathcal{I}[\hat{\theta}]\left[\delta[\hat{\theta}]\, P[\ell, m]\left[\mathrm{Cos}[\hat{\theta}]\right]\right]\right) \Big/ (m+\ell)!\right]\right] +$$
$$q\, \mathcal{S}[\ell]\left[r^{-1-\ell}\left(-a^{1+2\ell} + r^{1+2\ell}\right) \mathcal{I}_<[\hat{r}]\left[\hat{r}^{-1-\ell}\, \delta\left[\hat{r} - z_0\right]\right]\right.$$
$$\left.\mathcal{S}[m]\left[\left((-m+\ell)!\, \delta K[m, 0]\, P[\ell, m][\mathrm{Cos}[\theta]]\, \mathcal{I}[\hat{\theta}]\left[\delta[\hat{\theta}]\, P[\ell, m]\left[\mathrm{Cos}[\hat{\theta}]\right]\right]\right) \Big/ (m+\ell)!\right]\right]$$

where δK[m, 0] denotes a Kronecker δ.

Perform the $\hat{\theta}$ integration

```
w2[4] = w2[3] //. Op_[arg_][f_ δ[θ̂]] :> (f /. arg → 0)
```

$$q\, \mathcal{S}[\ell]\left[r^{-1-\ell}\, \mathcal{I}_>[\hat{r}]\left[\hat{r}^{-1-\ell}\left(-a^{1+2\ell} + \hat{r}^{1+2\ell}\right) \delta\left[\hat{r} - z_0\right]\right]\right.$$
$$\left.\mathcal{S}[m]\left[\left((-m+\ell)!\, \delta K[m, 0]\, P[\ell, m][1]\, P[\ell, m][\mathrm{Cos}[\theta]]\right) \Big/ (m+\ell)!\right]\right] +$$
$$q\, \mathcal{S}[\ell]\left[r^{-1-\ell}\left(-a^{1+2\ell} + r^{1+2\ell}\right) \mathcal{I}_<[\hat{r}]\left[\hat{r}^{-1-\ell}\, \delta\left[\hat{r} - z_0\right]\right]\right.$$
$$\left.\mathcal{S}[m]\left[\left((-m+\ell)!\, \delta K[m, 0]\, P[\ell, m][1]\, P[\ell, m][\mathrm{Cos}[\theta]]\right) \Big/ (m+\ell)!\right]\right]$$

Perform the $\hat{r}$ integration

```
w2[5] = w2[4] /. ℐ<[x_][f_. δ[x_ - a_]] :→ (f H[z0 - r] /. x → a) /.
  ℐ>[x_][f_. δ[x_ - a_]] :→ (f H[r - z0] /. x → a)
```

$$q\,H[-r+z_0]\,\mathcal{S}[\ell]\Big[r^{-1-\ell}\left(-a^{1+2\ell}+r^{1+2\ell}\right)z_0^{-1-\ell}\,\mathcal{S}[m]\Big[\left((-m+\ell)!\,\delta K[m,0]\,P[\ell,m][1]\,P[\ell,m][\mathrm{Cos}[\theta]]\right)\big/(m+\ell)!\Big]\Big]+q\,H[r-z_0]\,\mathcal{S}[\ell]\Big[r^{-1-\ell}z_0^{-1-\ell}\left(-a^{1+2\ell}+z_0^{1+2\ell}\right)\mathcal{S}[m]\Big[\left((-m+\ell)!\,\delta K[m,0]\,P[\ell,m][1]\,P[\ell,m][\mathrm{Cos}[\theta]]\right)\big/(m+\ell)!\Big]\Big]$$

Note the introduction of the Heaviside functions that restrict expressions to appropriate domains of validity.

Perform the summation over m

```
w2[6] = w2[5] //. Op_[m][f_ δK[m, a_]] :→ (f /. m → a)
```

$$q\,H[-r+z_0]\,\mathcal{S}[\ell]\left[r^{-1-\ell}\left(-a^{1+2\ell}+r^{1+2\ell}\right)z_0^{-1-\ell}P[\ell,0][1]\,P[\ell,0][\mathrm{Cos}[\theta]]\right]+q\,H[r-z_0]\,\mathcal{S}[\ell]\left[r^{-1-\ell}z_0^{-1-\ell}\left(-a^{1+2\ell}+z_0^{1+2\ell}\right)P[\ell,0][1]\,P[\ell,0][\mathrm{Cos}[\theta]]\right]$$

The associated Legendre polynomials are defined by $P_n^m(x) = (-1)^m\left(1-x^2\right)^{m/2}(d^m/dx^m)\,P_n(x)$ where P_n is the Legendre polynomial (LegendreP for Mathematica)

So

```
w2[7] = P[ℓ, 0][1] → LegendreP[ℓ, Cos[θ]] /. θ → 0
```

$$P[\ell,0][1]\to 1$$

```
w2[8] = w2[6] /. w2[7]
```

$$q\,H[-r+z_0]\,\mathcal{S}[\ell]\left[r^{-1-\ell}\left(-a^{1+2\ell}+r^{1+2\ell}\right)z_0^{-1-\ell}P[\ell,0][\mathrm{Cos}[\theta]]\right]+q\,H[r-z_0]\,\mathcal{S}[\ell]\left[r^{-1-\ell}z_0^{-1-\ell}\left(-a^{1+2\ell}+z_0^{1+2\ell}\right)P[\ell,0][\mathrm{Cos}[\theta]]\right]$$

I return to standard Mathematica notation

```
w2[9] =
 w2[8] /. a_ 𝒮[ℓ][b_] + c_ 𝒮[ℓ][d_] → Inactive[Sum][a b + c d, {ℓ, 0, ℓMax}] /.
  P[ℓ, 0][Cos[θ]] → LegendreP[ℓ, Cos[θ]]
```

$$\sum_{\ell=0}^{\ell\mathrm{Max}}\left(q\,r^{-1-\ell}z_0^{-1-\ell}\left(-a^{1+2\ell}+z_0^{1+2\ell}\right)H[r-z_0]\,\mathrm{LegendreP}[\ell,\mathrm{Cos}[\theta]]+q\,r^{-1-\ell}\left(-a^{1+2\ell}+r^{1+2\ell}\right)z_0^{-1-\ell}H[-r+\right.$$

I write a function for this expression.

```
Clear[ΦFromGreensFunction];
ΦFromGreensFunction[r_, θ_, z0 : _, q_, a_, ℓMax_] :=
```

$$\sum_{\ell=0}^{\ell Max} \left(q\, r^{-1-\ell}\, z_0^{-1-\ell} \left(-a^{1+2\ell} + z_0^{1+2\ell}\right) H[r - z_0]\, \text{LegendreP}[\ell, \text{Cos}[\theta]] + q\, r^{-1-\ell} \left(-a^{1+2\ell} + r^{1+2\ell}\right) z_0^{-1-\ell}\, H[-\right.$$

In notebook *Jackson Electrostatics - 1 09-16-17* the potential for this problem was obtained using the method of images.

```
w2[9] =
```

$$\Phi[r, \theta] == -\left((a\, q_p) \Big/ \left(r_p \sqrt{\left(r^2 + \frac{a^4}{r_p^2} - \frac{2\, a^2\, r\, \text{Cos}[\theta]}{r_p}\right)}\right)\right) + q_p \Big/ \left(\sqrt{\left(r^2 + r_p^2 - 2\, r\, r_p\, \text{Cos}[\theta]\right)}\right)$$

$$\Phi[r, \theta] == -\left((a\, q_p) \Big/ \left(\sqrt{\left(r^2 + \frac{a^4}{r_p^2} - \frac{2\, a^2\, r\, \text{Cos}[\theta]}{r_p}\right)}\, r_p\right)\right) + \frac{q_p}{\sqrt{r^2 - 2\, r\, \text{Cos}[\theta]\, r_p + r_p^2}}$$

I rewrite this to reflect the notation used in this notebook

```
w2[10] = w2[9] /. qp → q /. rp → z0
```

$$\Phi[r, \theta] == -\left((a\, q) \Big/ \left(z_0 \sqrt{\left(r^2 + \frac{a^4}{z_0^2} - \frac{2\, a^2\, r\, \text{Cos}[\theta]}{z_0}\right)}\right)\right) + \frac{q}{\sqrt{r^2 + z_0^2 - 2\, r\, z_0\, \text{Cos}[\theta]}}$$

To formally test the Green's function calculation, I could expand the method of images result, express it in terms of Legendre polynomial and then make term by term comparisons with the Green's function result. However, I will content myself with a numerical comparison for representative parameters and terms in the expansion (ℓMax = 30).

```
Clear[ΦFromMethodOfImages];
ΦFromMethodOfImages[r_, θ_, z0 : _, q_, a_] :=
```

$$-\left((a\, q) \Big/ \left(\sqrt{\left(r^2 + \frac{a^4}{z_0^2} - \frac{2\, a^2\, r\, \text{Cos}[\theta]}{z_0}\right)}\, z_0\right)\right) + \frac{q}{\sqrt{r^2 - 2\, r\, \text{Cos}[\theta]\, z_0 + z_0^2}}$$

```
fig1a = Module[{q = 1, a = 1, z0 = 1.75, conductor, pointCharge, range, lab, G},
   range = {{0, 3}, {0, 3}};
   lab = Stl["Φ from method of images"];
   conductor = {Yellow, Disk[{0, 0}, a]};
   pointCharge = {Blue,  Disk[{0, z0}, 0.02]};
   G[1] = Graphics[{conductor, pointCharge}, PlotRange → range, ImageSize → 250];
   G[2] = ContourPlot[ΦFromMethodOfImages[Sqrt[x^2 + z^2], ArcTan[x / z], z0, q, a], {x, 0,
      3}, {z, 0, 3}, FrameLabel → {{Rotate[Stl["z"], π/2], ""}, {Stl["x"], lab}},
     ColorFunction → "DeepSeaColors", Contours → 20, RegionFunction →
      Function[{x, z}, x^2 + z^2 > a^2], PlotRange → range, ImageSize → 250];
   Show[
    G[
     2],
    G[
     1]]];
```

```
fig1b =
  Module[{q = 1, a = 1, z0 = 1.75, ℓMax = 30, conductor, pointCharge, range, lab, G},
   range = {{0, 3}, {0, 3}};
   lab = Stl["Φ from Green's function"];
   conductor = {Yellow, Disk[{0, 0}, a]};
   pointCharge = {Blue,  Disk[{0, z0}, 0.02]};
   G[1] = Graphics[{conductor, pointCharge}, PlotRange → range, ImageSize → 250];
   G[2] = ContourPlot[Activate@
      ΦFromGreensFunction[Sqrt[x^2 + z^2], ArcTan[x / z], z0, q, a, ℓMax], {x, 0, 3},
     {z, 0, 3}, FrameLabel → {{Rotate[Stl["z"], π/2], ""}, {Stl["x"], lab}},
     ColorFunction → "DeepSeaColors", Contours → 20, RegionFunction →
      Function[{x, z}, x^2 + z^2 > a^2], PlotRange → range, ImageSize → 250];
   Show[
    G[
     2],
    G[
     1]]];
```

```
Grid[{{fig1a, fig1b}}]
```



3 Conducting sphere with external ring of charge



I calculate the potential associated with a charged ring outside a conducting sphere with radius *a*.

Without loss of generality, the ring charge is chosen to lie in the x-y plane, and the problem is axisymmet ric since the charge distribution has no explicit ϕ-dependence

In this case, the charged ring is described by

```
Symbolize[ r_c ];
```

$$\text{def}[\rho\text{Ring}] = \rho[\hat{r}, \hat{\theta}, \hat{\phi}] \to q \frac{\delta[\hat{r} - r_c]\, \delta[\hat{\theta} - \frac{\pi}{2}]}{2\pi \hat{r}^2 \,\text{Sin}[\hat{\theta}]}$$

$$\rho[\hat{r}, \hat{\theta}, \hat{\phi}] \to \frac{1}{\pi \hat{r}^2} q\, \text{Csc}[\hat{\theta}]\, \delta[\hat{r} - r_c]\, \delta[-\pi + 2\hat{\theta}]$$

I check the normalization

$$\text{Integrate}\big[\hat{r}^2\, \text{Integrate}\big[\text{Sin}[\hat{\theta}]\, \text{Integrate}\big[(\rho[\hat{r}, \hat{\theta}, \hat{\phi}] \text{ /. def}[\rho\text{Ring}]), \{\hat{\phi}, 0, 2\pi\}\big], \{\hat{\theta}, 0, \pi\}\big], \{\hat{r}, 0, \infty\}, \text{Assumptions} \to r_c > 0\big]$$

q

For this problem, equation (4) for Φ(r, θ) is represented by

$$\text{w3}[1] = \mathcal{S}[\ell]\big[\mathcal{S}[m]\big[\mathcal{I}[\hat{\phi}]\big[\mathcal{I}[\hat{\theta}]\big[\text{Sin}[\hat{\theta}]\, \mathcal{I}_<[\hat{r}]\big[\hat{r}^2 \rho[\hat{r}, \hat{\theta}, \hat{\phi}]\, G_<[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi]\big]\big]\big]\big]\big] + \mathcal{S}[\ell]\big[\mathcal{S}[m]\big[\mathcal{I}[\hat{\phi}]\big[\mathcal{I}[\hat{\theta}]\big[\text{Sin}[\hat{\theta}]\, \mathcal{I}_>[\hat{r}]\big[\hat{r}^2 \rho[\hat{r}, \hat{\theta}, \hat{\phi}]\, G_>[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi]\big]\big]\big]\big]\big]$$

$$\mathcal{S}[\ell]\big[\mathcal{S}[m]\big[\mathcal{I}[\hat{\phi}]\big[\mathcal{I}[\hat{\theta}]\big[\text{Sin}[\hat{\theta}]\, \mathcal{I}_>[\hat{r}]\big[\hat{r}^2 G_>[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi]\, \rho[\hat{r}, \hat{\theta}, \hat{\phi}]\big]\big]\big]\big]\big] + \mathcal{S}[\ell]\big[\mathcal{S}[m]\big[\mathcal{I}[\hat{\phi}]\big[\mathcal{I}[\hat{\theta}]\big[\text{Sin}[\hat{\theta}]\, \mathcal{I}_<[\hat{r}]\big[\hat{r}^2 G_<[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi]\, \rho[\hat{r}, \hat{\theta}, \hat{\phi}]\big]\big]\big]\big]\big]$$

The charged ring is outside the conductor. Like the previous example, the appropriate radial component of the Green's function is (derived above as w1["no outer conductor"])

$$\text{w1}[\text{"no outer conductor"}] = \big\{g_<[r, \hat{r}] == \big(4\pi r^{-1-\ell}\,(-a^{1+2\ell} + r^{1+2\ell})\, \hat{r}^{-1-\ell}\big) / (1 + 2\ell),\; g_>[r, \hat{r}] == \big(4\pi r^{-1-\ell}\, \hat{r}^{-1-\ell}\,(-a^{1+2\ell} + \hat{r}^{1+2\ell})\big) / (1 + 2\ell)\big\}$$

$$\Big\{g_<[r, \hat{r}] == \frac{1}{1+2\ell} 4\pi r^{-1-\ell}\,(-a^{1+2\ell} + r^{1+2\ell})\, \hat{r}^{-1-\ell},\; g_>[r, \hat{r}] == \frac{1}{1+2\ell} 4\pi r^{-1-\ell}\, \hat{r}^{-1-\ell}\,(-a^{1+2\ell} + \hat{r}^{1+2\ell})\Big\}$$

Then

```
def[G<] = G<[ℓ, m, r̂, θ̂, ϕ̂, r, θ, ϕ] → g<[r, r̂] Y[ℓ, m][θ, ϕ] Ys[ℓ, m][θ̂, ϕ̂] /.
```
$$\left\{Y[\ell, m][\theta, \phi] \to \sqrt{\frac{2\ell+1}{4\pi}}\sqrt{\frac{(\ell-m)!}{(\ell+m)!}}\, P[\ell, m][\mathrm{Cos}[\theta]]\, \mathrm{Exp}[I\, m\, \phi],\right.$$
$$\left.Ys[\ell, m][\hat{\theta}, \hat{\phi}] \to \sqrt{\frac{2\ell+1}{4\pi}}\sqrt{\frac{(\ell-m)!}{(\ell+m)!}}\, P[\ell, m][\mathrm{Cos}[\hat{\theta}]]\, \mathrm{Exp}[-I\, m\, \hat{\phi}]\right\} \ /.$$
```
(w1["no outer conductor"] // ER)
```

$$G_<[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi] \to \left(e^{i m \phi - i m \hat{\phi}}\, r^{-1-\ell}\left(-a^{1+2\ell} + r^{1+2\ell}\right)\hat{r}^{-1-\ell}(-m+\ell)!\, P[\ell, m][\mathrm{Cos}[\theta]]\, P[\ell, m][\mathrm{Cos}[\hat{\theta}]]\right)\Big/(m+\ell)!$$

```
def[G>] = G>[ℓ, m, r̂, θ̂, ϕ̂, r, θ, ϕ] → g>[r, r̂] Y[ℓ, m][θ, ϕ] Ys[ℓ, m][θ̂, ϕ̂] /.
```
$$\left\{Y[\ell, m][\theta, \phi] \to \sqrt{\frac{2\ell+1}{4\pi}}\sqrt{\frac{(\ell-m)!}{(\ell+m)!}}\, P[\ell, m][\mathrm{Cos}[\theta]]\, \mathrm{Exp}[I\, m\, \phi],\right.$$
$$\left.Ys[\ell, m][\hat{\theta}, \hat{\phi}] \to \sqrt{\frac{2\ell+1}{4\pi}}\sqrt{\frac{(\ell-m)!}{(\ell+m)!}}\, P[\ell, m][\mathrm{Cos}[\hat{\theta}]]\, \mathrm{Exp}[-I\, m\, \hat{\phi}]\right\} \ /.$$
```
(w1["no outer conductor"] // ER)
```

$$G_>[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi] \to \left(e^{i m \phi - i m \hat{\phi}}\, r^{-1-\ell}\,\hat{r}^{-1-\ell}\left(-a^{1+2\ell} + \hat{r}^{1+2\ell}\right)(-m+\ell)!\, P[\ell, m][\mathrm{Cos}[\theta]]\, P[\ell, m][\mathrm{Cos}[\hat{\theta}]]\right)\Big/(m+\ell)!$$

Using the explicit forms for the charge density and Green’s function

```
w3[2] = w3[1] //. def[ρRing] /. def[G<] /. def[G>]
```

$$\frac{1}{\pi} q\, \mathcal{S}[\ell]\Big[r^{-1-\ell}\, \mathcal{I}_>[\hat{r}]\Big[\hat{r}^{-1-\ell}\left(-a^{1+2\ell}+\hat{r}^{1+2\ell}\right)\delta[\hat{r}-r_c]\Big]\, \mathcal{S}[m]\Big[\Big((-m+\ell)!\, P[\ell, m][\mathrm{Cos}[\theta]]\, \mathcal{I}[\hat{\theta}]\Big[\delta[-\pi+2\hat{\theta}]\, P[\ell, m][\mathrm{Cos}[\hat{\theta}]]\Big]\, \mathcal{I}[\hat{\phi}]\Big[e^{i m \phi - i m \hat{\phi}}\Big]\Big)\Big/(m+\ell)!\Big]\Big] +$$
$$\frac{1}{\pi} q\, \mathcal{S}[\ell]\Big[r^{-1-\ell}\left(-a^{1+2\ell}+r^{1+2\ell}\right)\mathcal{I}_<[\hat{r}]\Big[\hat{r}^{-1-\ell}\,\delta[\hat{r}-r_c]\Big]\, \mathcal{S}[m]\Big[\Big((-m+\ell)!\, P[\ell, m][\mathrm{Cos}[\theta]]\, \mathcal{I}[\hat{\theta}]\Big[\delta[-\pi+2\hat{\theta}]\, P[\ell, m][\mathrm{Cos}[\hat{\theta}]]\Big]\, \mathcal{I}[\hat{\phi}]\Big[e^{i m \phi - i m \hat{\phi}}\Big]\Big)\Big/(m+\ell)!\Big]\Big]$$

Perform the $\hat{\phi}$ integration

```
w3[3] = w3[2] //. Op_[ϕ̂][f_] :> (2 π δK[m, 0])
```

$$2 q\, \mathcal{S}[\ell]\Big[r^{-1-\ell}\, \mathcal{I}_>[\hat{r}]\Big[\hat{r}^{-1-\ell}\left(-a^{1+2\ell}+\hat{r}^{1+2\ell}\right)\delta[\hat{r}-r_c]\Big]\, \mathcal{S}[m]\Big[\Big((-m+\ell)!\, \delta K[m, 0]\, P[\ell, m][\mathrm{Cos}[\theta]]\, \mathcal{I}[\hat{\theta}]\Big[\delta[-\pi+2\hat{\theta}]\, P[\ell, m][\mathrm{Cos}[\hat{\theta}]]\Big]\Big)\Big/(m+\ell)!\Big]\Big] +$$
$$2 q\, \mathcal{S}[\ell]\Big[r^{-1-\ell}\left(-a^{1+2\ell}+r^{1+2\ell}\right)\mathcal{I}_<[\hat{r}]\Big[\hat{r}^{-1-\ell}\,\delta[\hat{r}-r_c]\Big]\, \mathcal{S}[m]\Big[\Big((-m+\ell)!\, \delta K[m, 0]\, P[\ell, m][\mathrm{Cos}[\theta]]\, \mathcal{I}[\hat{\theta}]\Big[\delta[-\pi+2\hat{\theta}]\, P[\ell, m][\mathrm{Cos}[\hat{\theta}]]\Big]\Big)\Big/(m+\ell)!\Big]\Big]$$

where δK[m, 0] denotes a Kronecker δ.

Perform the $\hat{\theta}$ integration

```
w3[4] = w3[3] //. Op_[arg_][f_ δ[-π + 2 θ̂]] :→ (f/2 /. arg → π/2)
```

$$q\,\mathcal{S}[\ell]\Big[r^{-1-\ell}\,\mathcal{I}_{>}[\hat{r}]\Big[\hat{r}^{-1-\ell}\left(-a^{1+2\ell}+\hat{r}^{1+2\ell}\right)\delta[\hat{r}-r_c]\Big]\;\mathcal{S}[m]\Big[\big((-m+\ell)!\,\delta K[m,0]\,P[\ell,m][0]\,P[\ell,m][\mathrm{Cos}[\Theta]]\big)\big/(m+\ell)!\Big]\Big]+$$
$$q\,\mathcal{S}[\ell]\Big[r^{-1-\ell}\left(-a^{1+2\ell}+r^{1+2\ell}\right)\mathcal{I}_{<}[\hat{r}]\Big[\hat{r}^{-1-\ell}\,\delta[\hat{r}-r_c]\Big]\;\mathcal{S}[m]\Big[\big((-m+\ell)!\,\delta K[m,0]\,P[\ell,m][0]\,P[\ell,m][\mathrm{Cos}[\Theta]]\big)\big/(m+\ell)!\Big]\Big]$$

where it should be noted that I had to fiddle with the δ-function integral rule because the Mathematica evaluator expanded the original form $\delta(\hat{\theta} - \pi/2)$ into $\delta\left[-\pi + 2\,\hat{\theta}\right]/2$

Perform the $\hat{r}$ integration

```
w3[5] = w3[4] /. 𝓘<[x_][f_. δ[x_ - a_]] :→ (f H[rc - r] /. x → a) /.
  𝓘>[x_][f_. δ[x_ - a_]] :→ (f H[r - rc] /. x → a)
```

$$q\,H[-r+r_c]\,\mathcal{S}[\ell]\Big[r^{-1-\ell}\left(-a^{1+2\ell}+r^{1+2\ell}\right)r_c^{-1-\ell}\;\mathcal{S}[m]\Big[\big((-m+\ell)!\,\delta K[m,0]\,P[\ell,m][0]\,P[\ell,m][\mathrm{Cos}[\Theta]]\big)\big/(m+\ell)!\Big]\Big]+$$
$$q\,H[r-r_c]\,\mathcal{S}[\ell]\Big[r^{-1-\ell}\,r_c^{-1-\ell}\left(-a^{1+2\ell}+r_c^{1+2\ell}\right)\;\mathcal{S}[m]\Big[\big((-m+\ell)!\,\delta K[m,0]\,P[\ell,m][0]\,P[\ell,m][\mathrm{Cos}[\Theta]]\big)\big/(m+\ell)!\Big]\Big]$$

Note the introduction of the Heaviside functions that restrict expressions to appropriate domains of validity.

Perform the summation over m

```
w3[6] = w3[5] //. Op_[m][f_ δK[m, a_]] :→ (f /. m → a)
```

$$q\,H[-r+r_c]\,\mathcal{S}[\ell]\Big[r^{-1-\ell}\left(-a^{1+2\ell}+r^{1+2\ell}\right)r_c^{-1-\ell}\,P[\ell,0][0]\,P[\ell,0][\mathrm{Cos}[\Theta]]\Big]+$$
$$q\,H[r-r_c]\,\mathcal{S}[\ell]\Big[r^{-1-\ell}\,r_c^{-1-\ell}\left(-a^{1+2\ell}+r_c^{1+2\ell}\right)P[\ell,0][0]\,P[\ell,0][\mathrm{Cos}[\Theta]]\Big]$$

which is the result Jackson 3.130 appropriate for r > r_c (using a slightly different notation).

I return to standard Mathematica form

```
w3[7] =
 w3[6] /. a_ 𝓢[ℓ][b_] + c_ 𝓢[ℓ][d_] → Inactive[Sum][a b + c d, {ℓ, 0, ℓMax}] /.
  P[index_, 0][arg_] → LegendreP[index, arg]
```

$$\sum_{\ell=0}^{\ell\mathrm{Max}}\left(q\,r^{-1-\ell}\,r_c^{-1-\ell}\left(-a^{1+2\ell}+r_c^{1+2\ell}\right)H[r-r_c]\,\mathrm{LegendreP}[\ell,0]\,\mathrm{LegendreP}[\ell,\mathrm{Cos}[\Theta]]+q\,r^{-1-\ell}\left(-a^{1+2\ell}+\right.\right.$$

I write a function for this expression.

```
Clear[ΦRingGreensFunction];
ΦRingGreensFunction[r_, θ_, r_c : _, q_, a_, ℓMax_] :=
 Sum_{ℓ=0}^{ℓMax} (q r^(-1-ℓ) r_c^(-1-ℓ) (-a^(1+2ℓ) + r_c^(1+2ℓ)) H[r - r_c] LegendreP[ℓ, 0] LegendreP[ℓ, Cos[θ]] + q r^(-1-ℓ) (-a^(1+2ℓ)
```

For example

```
Activate@ΦRingGreensFunction[r, θ, z0, q, a, 2]
```

$$\frac{q\,(-a+z_0)\,H[r-z_0]}{r\,z_0}-\frac{1}{4\,r^3\,z_0^3}q\,\left(-a^5+z_0^5\right)\,\left(-1+3\,\mathrm{Cos}[\theta]^2\right)\,H[r-z_0]+$$
$$\frac{q\,(-a+r)\,H[-r+z_0]}{r\,z_0}-\frac{1}{4\,r^3\,z_0^3}q\,\left(-a^5+r^5\right)\,\left(-1+3\,\mathrm{Cos}[\theta]^2\right)\,H[-r+z_0]$$

For representative parameters and terms in the expansion (ℓMax = 30).

```
Module[{q = 1, a = 1, r_c = 1.75, ℓMax = 30, conductor, ringCharge, range, lab, G},
 range = {{0, 3}, {0, 3}};
 lab = Stl["Φ from Green's function"];
 conductor = {Yellow, Disk[{0, 0}, a]};
 ringCharge = {Blue, Disk[{r_c, 0}, 0.02]};
 G[1] = Graphics[{conductor, ringCharge}, PlotRange → range, ImageSize → 250];
 G[2] = ContourPlot[
   Activate@ΦRingGreensFunction[Sqrt[x^2 + z^2], ArcTan[x / z], r_c, q, a, ℓMax], {x, 0, 3},
   {z, 0, 3}, FrameLabel → {{Rotate[Stl["z"], π/2], ""}, {Stl["x"], lab}},
   ColorFunction → "DeepSeaColors", Contours → 20, RegionFunction →
    Function[{x, z}, x^2 + z^2 > a^2], PlotRange → range, ImageSize → 250];
 Show[
  G[
   2],
  G[
   1]]]
```



4 Conducting sphere with internal line charge



I calculate the potential associated with a line charge inside a conducting sphere with radius *b*.

Without loss of generality, the line charge is oriented along the z-axis, and the problem is axisymmetric since the charge distribution has no explicit ϕ-dependence

Assuming the linear charge density is constant, the line of charge is described by

$$\text{def}[\rho\text{Line}] = \rho[\hat{r}, \hat{\theta}, \hat{\phi}] \rightarrow \frac{q}{2\,b} \frac{1}{2\,\pi\,\hat{r}^2} (\delta[\text{Cos}[\hat{\theta}] - 1] + \delta[\text{Cos}[\hat{\theta}] + 1])$$

$$\rho[\hat{r}, \hat{\theta}, \hat{\phi}] \rightarrow (q\,(\delta[-1 + \text{Cos}[\hat{\theta}]] + \delta[1 + \text{Cos}[\hat{\theta}]]))\,/\,(4\,b\,\pi\,\hat{r}^2)$$

Check the normalization

$$\text{normCheck}[1] = \text{Int}[\text{Sin}[\hat{\theta}]\ \text{Integrate}[\hat{r}^2\ \text{Integrate}[(\rho[\hat{r}, \hat{\theta}, \hat{\phi}]\ /.\ \text{def}[\rho\text{Line}]), \{\hat{\phi}, 0, 2\,\pi\}], \{\hat{r}, 0, b\}], \{\hat{\theta}, 0, \pi\}]$$

$$\text{Int}\left[\frac{1}{2}\,q\,(\delta[-1 + \text{Cos}[\hat{\theta}]] + \delta[1 + \text{Cos}[\hat{\theta}]])\ \text{Sin}[\hat{\theta}], \{\hat{\theta}, 0, \pi\}\right]$$

Mathematica does not correctly handle this integral

```
normCheck[1] /. Int → Integrate
```

```
0
```

I make progress by changing variables

$$x = \cos(\hat{\theta}) \;\Rightarrow\; \mathrm{dx} = -\sin(\hat{\theta})\, d\,\hat{\theta}$$
$$\hat{\theta} = 0 \;\Rightarrow\; x = 1$$

$$\hat{\theta} = \pi \;\Rightarrow\; x = -1$$
$$\mathcal{J} = \frac{q}{2}\int_0^{\pi} \mathrm{Sin}[\hat{\theta}]\left(\delta\left[-1 + \mathrm{Cos}[\hat{\theta}]\right] + \delta\left[1 + \mathrm{Cos}[\hat{\theta}]\right]\right) d\hat{\theta}$$
$$= \frac{q}{2}\int_1^{-1} (-\mathrm{dx})\,(\delta[x-1] + \delta[x+1])\, dx$$
$$= \frac{q}{2}\int_{-1}^{1} \mathrm{dx}\,(\delta[x-1] + \delta[x+1])\, dx$$
$$= \frac{q}{2}\,2$$
$$= q$$

For this problem, equation (4) for $\Phi(\mathrm{r}, \theta)$ is represented by

```
w4[1] =
  𝒮[ℓ] [𝒮[m] [ ℐ[φ̂] [ℐ[θ̂] [Sin[θ̂] ℐ<[r̂] [r̂² ρ[r̂, θ̂, φ̂] G<[ℓ, m, r̂, θ̂, φ̂, r, θ, φ]]]] ]] +
   𝒮[ℓ] [𝒮[m] [ ℐ[φ̂] [ℐ[θ̂] [Sin[θ̂] ℐ>[r̂] [r̂² ρ[r̂, θ̂, φ̂] G>[ℓ, m, r̂, θ̂, φ̂, r, θ, φ]]]] ]]
```

```
𝒮[ℓ] [𝒮[m] [ℐ[φ̂] [ℐ[θ̂] [Sin[θ̂] ℐ>[r̂] [r̂² G>[ℓ, m, r̂, θ̂, φ̂, r, θ, φ] ρ[r̂, θ̂, φ̂]]]]]] +
 𝒮[ℓ] [𝒮[m] [ℐ[φ̂] [ℐ[θ̂] [Sin[θ̂] ℐ<[r̂] [r̂² G<[ℓ, m, r̂, θ̂, φ̂, r, θ, φ] ρ[r̂, θ̂, φ̂]]]]]]
```

The charged ring is inside the conductor.The appropriate radial component of the Green's function is (derived above as w1["no inner conductor"])

```
w1["no inner conductor"] = {g<[r, r̂] == (4 π r^ℓ r̂^-ℓ (1/r̂ - b^(-1-2ℓ) r̂^(2ℓ))) / (1 + 2 ℓ),
   g>[r, r̂] == (4 π r^-ℓ (1/r - b^(-1-2ℓ) r^(2ℓ)) r̂^ℓ) / (1 + 2 ℓ)}
```

$$\left\{g_<[\mathrm{r}, \hat{\mathrm{r}}] == \frac{4\pi\, \mathrm{r}^{\ell}\, \hat{\mathrm{r}}^{-\ell}\left(\frac{1}{\hat{\mathrm{r}}} - \mathrm{b}^{-1-2\ell}\,\hat{\mathrm{r}}^{2\ell}\right)}{1 + 2\ell},\; g_>[\mathrm{r}, \hat{\mathrm{r}}] == \frac{4\pi\, \mathrm{r}^{-\ell}\left(\frac{1}{\mathrm{r}} - \mathrm{b}^{-1-2\ell}\,\mathrm{r}^{2\ell}\right)\hat{\mathrm{r}}^{\ell}}{1 + 2\ell}\right\}$$

Then

$$\text{def}[G_<] = G_<[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi] \to g_<[r, \hat{r}]\, Y[\ell, m][\theta, \phi]\, Ys[\ell, m][\hat{\theta}, \hat{\phi}] \;/.$$
$$\left\{Y[\ell, m][\theta, \phi] \to \sqrt{\frac{2\ell+1}{4\pi}} \sqrt{\frac{(\ell-m)!}{(\ell+m)!}}\, P[\ell, m][\text{Cos}[\theta]]\, \text{Exp}[I\, m\, \phi],\right.$$
$$\left. Ys[\ell, m][\hat{\theta}, \hat{\phi}] \to \sqrt{\frac{2\ell+1}{4\pi}} \sqrt{\frac{(\ell-m)!}{(\ell+m)!}}\, P[\ell, m][\text{Cos}[\hat{\theta}]]\, \text{Exp}[-I\, m\, \hat{\phi}]\right\} \;/.$$
$$(\text{w1}[\text{"no inner conductor"}] \;//\; \text{ER})$$

$$G_<[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi] \to \left(e^{i m \phi - i m \hat{\phi}}\, r^{\ell}\, \hat{r}^{-\ell} \left(\frac{1}{\hat{r}} - b^{-1-2\ell}\, \hat{r}^{2\ell}\right) (-m+\ell)!\, P[\ell, m][\text{Cos}[\theta]]\, P[\ell, m][\text{Cos}[\hat{\theta}]]\right) \Big/ (m+\ell)!$$

$$\text{def}[G_>] = G_>[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi] \to g_>[r, \hat{r}]\, Y[\ell, m][\theta, \phi]\, Ys[\ell, m][\hat{\theta}, \hat{\phi}] \;/.$$
$$\left\{Y[\ell, m][\theta, \phi] \to \sqrt{\frac{2\ell+1}{4\pi}} \sqrt{\frac{(\ell-m)!}{(\ell+m)!}}\, P[\ell, m][\text{Cos}[\theta]]\, \text{Exp}[I\, m\, \phi],\right.$$
$$\left. Ys[\ell, m][\hat{\theta}, \hat{\phi}] \to \sqrt{\frac{2\ell+1}{4\pi}} \sqrt{\frac{(\ell-m)!}{(\ell+m)!}}\, P[\ell, m][\text{Cos}[\hat{\theta}]]\, \text{Exp}[-I\, m\, \hat{\phi}]\right\} \;/.$$
$$(\text{w1}[\text{"no inner conductor"}] \;//\; \text{ER})$$

$$G_>[\ell, m, \hat{r}, \hat{\theta}, \hat{\phi}, r, \theta, \phi] \to \left(e^{i m \phi - i m \hat{\phi}}\, r^{-\ell} \left(\frac{1}{r} - b^{-1-2\ell}\, r^{2\ell}\right) \hat{r}^{\ell}\, (-m+\ell)!\, P[\ell, m][\text{Cos}[\theta]]\, P[\ell, m][\text{Cos}[\hat{\theta}]]\right) \Big/ (m+\ell)!$$

Using the explicit forms for the charge density and Green's function

$$\text{w4}[2] = \text{w4}[1] \;//.\; \text{def}[\rho\text{Line}] \;/.\; \text{def}[G_<] \;/.\; \text{def}[G_>]$$

$$\frac{1}{4 b \pi} q\, \mathcal{S}[\ell]\Big[r^{-\ell}\left(\frac{1}{r} - b^{-1-2\ell}\, r^{2\ell}\right) \mathcal{I}_>[\hat{r}][\hat{r}^{\ell}]\, \mathcal{S}[m]\Big[\frac{1}{(m+\ell)!}(-m+\ell)!\, P[\ell, m][\text{Cos}[\theta]]$$
$$\mathcal{I}[\hat{\theta}]\Big[\left(\delta[-1+\text{Cos}[\hat{\theta}]] + \delta[1+\text{Cos}[\hat{\theta}]]\right) \text{Sin}[\hat{\theta}]\, P[\ell, m][\text{Cos}[\hat{\theta}]]\Big]\, \mathcal{I}[\hat{\phi}]\Big[e^{i m \phi - i m \hat{\phi}}\Big]\Big]\Big] +$$
$$\frac{1}{4 b \pi} q\, \mathcal{S}[\ell]\Big[r^{\ell}\, \mathcal{I}_<[\hat{r}]\Big[\hat{r}^{-\ell}\left(\frac{1}{\hat{r}} - b^{-1-2\ell}\, \hat{r}^{2\ell}\right)\Big]\, \mathcal{S}[m]\Big[\frac{1}{(m+\ell)!}(-m+\ell)!\, P[\ell, m][\text{Cos}[\theta]]$$
$$\mathcal{I}[\hat{\theta}]\Big[\left(\delta[-1+\text{Cos}[\hat{\theta}]] + \delta[1+\text{Cos}[\hat{\theta}]]\right) \text{Sin}[\hat{\theta}]\, P[\ell, m][\text{Cos}[\hat{\theta}]]\Big]\, \mathcal{I}[\hat{\phi}]\Big[e^{i m \phi - i m \hat{\phi}}\Big]\Big]\Big]$$

Perform the $\hat{\phi}$ integration

```
w4[3] = w4[2] //. Op_[φ̂][f_] :> (2 π δK[m, 0])
```

$$\frac{1}{2\,b} q\,\mathcal{S}[\ell]\Big[r^{-\ell}\left(\frac{1}{r}-b^{-1-2\ell}r^{2\ell}\right)\mathcal{I}_{>}[\hat{r}][\hat{r}^{\ell}]\,\mathcal{S}[m]\Big[\frac{1}{(m+\ell)!}(-m+\ell)!\,\delta K[m,0]\,P[\ell,m][\mathrm{Cos}[\theta]]\,\mathcal{I}[\hat{\theta}]\Big[\left(\delta[-1+\mathrm{Cos}[\hat{\theta}]]+\delta[1+\mathrm{Cos}[\hat{\theta}]]\right)\mathrm{Sin}[\hat{\theta}]\,P[\ell,m][\mathrm{Cos}[\hat{\theta}]]\Big]\Big]\Big]+\frac{1}{2\,b} q\,\mathcal{S}[\ell]\Big[r^{\ell}\,\mathcal{I}_{<}[\hat{r}]\Big[\hat{r}^{-\ell}\left(\frac{1}{\hat{r}}-b^{-1-2\ell}\hat{r}^{2\ell}\right)\Big]\,\mathcal{S}[m]\Big[\frac{1}{(m+\ell)!}(-m+\ell)!\,\delta K[m,0]\,P[\ell,m][\mathrm{Cos}[\theta]]\,\mathcal{I}[\hat{\theta}]\Big[\left(\delta[-1+\mathrm{Cos}[\hat{\theta}]]+\delta[1+\mathrm{Cos}[\hat{\theta}]]\right)\mathrm{Sin}[\hat{\theta}]\,P[\ell,m][\mathrm{Cos}[\hat{\theta}]]\Big]\Big]\Big]$$

where δK[m, 0] denotes a Kronecker δ.

Perform the $\hat{\theta}$ integration

It is easier for Mathematica to perform the $\hat{\theta}$ integration if the change of variables cos($\hat{\theta}$) → x is made

$$\mathcal{I}[\hat{\theta}][f] \equiv \int_0^{\pi} \sin(\hat{\theta})\, f(\hat{\theta})\, d\hat{\theta} \to \int_{-1}^{1} f(x)\, dx$$

```
w4[4] = w4[3] //. I[θ̂][(δ[-1 + Cos[θ̂]] + δ[1 + Cos[θ̂]]) Sin[θ̂] P[ℓ, m][Cos[θ̂]]] →
    I[x][δ[x - 1] P[ℓ, 0][x]] + I[x][δ[x + 1] P[ℓ, 0][x]]
```

$$\frac{1}{2\,b} q\,\mathcal{S}[\ell]\Big[r^{-\ell}\left(\frac{1}{r}-b^{-1-2\ell}r^{2\ell}\right)\left(\mathcal{I}[x][\delta[-1+x]\,P[\ell,0][x]]+\mathcal{I}[x][\delta[1+x]\,P[\ell,0][x]]\right)\mathcal{I}_{>}[\hat{r}][\hat{r}^{\ell}]\,\mathcal{S}[m]\Big[\left((-m+\ell)!\,\delta K[m,0]\,P[\ell,m][\mathrm{Cos}[\theta]]\right)\Big/(m+\ell)!\Big]\Big]+\frac{1}{2\,b} q\,\mathcal{S}[\ell]\Big[r^{\ell}\left(\mathcal{I}[x][\delta[-1+x]\,P[\ell,0][x]]+\mathcal{I}[x][\delta[1+x]\,P[\ell,0][x]]\right)\mathcal{I}_{<}[\hat{r}]\Big[\hat{r}^{-\ell}\left(\frac{1}{\hat{r}}-b^{-1-2\ell}\hat{r}^{2\ell}\right)\Big]\,\mathcal{S}[m]\Big[\left((-m+\ell)!\,\delta K[m,0]\,P[\ell,m][\mathrm{Cos}[\theta]]\right)\Big/(m+\ell)!\Big]\Big]$$

The rule for the x-integral is easily implemented

```
w4[5] = w4[4] //. I[x_][f_. δ[x_ + a_]] :> (f /. x → -a)
```

$$\frac{1}{2\,b} q\,\mathcal{S}[\ell]\Big[r^{-\ell}\left(\frac{1}{r}-b^{-1-2\ell}r^{2\ell}\right)\left(P[\ell,0][-1]+P[\ell,0][1]\right)\mathcal{I}_{>}[\hat{r}][\hat{r}^{\ell}]\,\mathcal{S}[m]\Big[\left((-m+\ell)!\,\delta K[m,0]\,P[\ell,m][\mathrm{Cos}[\theta]]\right)\Big/(m+\ell)!\Big]\Big]+\frac{1}{2\,b} q\,\mathcal{S}[\ell]\Big[r^{\ell}\left(P[\ell,0][-1]+P[\ell,0][1]\right)\mathcal{I}_{<}[\hat{r}]\Big[\hat{r}^{-\ell}\left(\frac{1}{\hat{r}}-b^{-1-2\ell}\hat{r}^{2\ell}\right)\Big]\,\mathcal{S}[m]\Big[\left((-m+\ell)!\,\delta K[m,0]\,P[\ell,m][\mathrm{Cos}[\theta]]\right)\Big/(m+\ell)!\Big]\Big]$$

Perform the summation over m

```
w4[6] = w4[5] //. Op_[m][f_ δK[m, a_]] :> (f /. m → a)
```

$$\frac{1}{2\,b} q\,\mathcal{S}[\ell]\left[r^{-\ell}\left(\frac{1}{r} - b^{-1-2\ell}\,r^{2\ell}\right)\,(P[\ell,0][-1] + P[\ell,0][1])\,P[\ell,0][\mathrm{Cos}[\theta]]\,\mathcal{I}_{>}[\hat{r}][\hat{r}^{\ell}]\right] + \frac{1}{2\,b} q\,\mathcal{S}[\ell]\left[r^{\ell}\,(P[\ell,0][-1] + P[\ell,0][1])\,P[\ell,0][\mathrm{Cos}[\theta]]\,\mathcal{I}_{<}[\hat{r}]\left[\hat{r}^{-\ell}\left(\frac{1}{\hat{r}} - b^{-1-2\ell}\,\hat{r}^{2\ell}\right)\right]\right]$$

The $\hat{r}$ integrations are performed with

```
w4[7] = w4[6] /.
    𝓘<[r̂][a_] :> Integrate[a, {r̂, r, b}, Assumptions → {r > 0, b > r, ℓ ≥ 0}] /.
   𝓘>[r̂][a_] :> Integrate[a, {r̂, 0, r}, Assumptions → {r > 0, b > r, ℓ ≥ 0}]
```

$$\frac{1}{2\,b} q\,r\,\mathcal{S}[\ell]\left[\frac{1}{1+\ell}\left(\frac{1}{r} - b^{-1-2\ell}\,r^{2\ell}\right)\,(P[\ell,0][-1] + P[\ell,0][1])\,P[\ell,0][\mathrm{Cos}[\theta]]\right] + \frac{1}{2\,b} q\,\mathcal{S}[\ell]\left[r^{\ell}\left(\frac{-b^{-\ell} + r^{-\ell}}{\ell} + \frac{b^{-1-2\ell}\,(-b^{1+\ell} + r^{1+\ell})}{1+\ell}\right)\,(P[\ell,0][-1] + P[\ell,0][1])\,P[\ell,0][\mathrm{Cos}[\theta]]\right]$$

Some pattern matching can simplify this expression.

```
w4[8] = w4[7] /. a_ 𝓢[ℓ][b_] + c_ 𝓢[ℓ][d_] :> temp[ℓ][Simplify[a b + c d]]  /. temp → 𝓢
```

$$\frac{1}{2} q\,\mathcal{S}[\ell]\left[\left(b^{-1-\ell}\,(b^{\ell} - r^{\ell})\,(1+2\,\ell)\,(P[\ell,0][-1] + P[\ell,0][1])\,P[\ell,0][\mathrm{Cos}[\theta]]\right) / (\ell\,(1+\ell))\right]$$

Returning to standard Mathematica functions

```
w4[9] = w4[8]  /. a_ 𝓢[ℓ][b_] → Inactive[Sum][a b , {ℓ, 0, ℓMax}] /.
   P[index_, 0][arg_] → LegendreP[index, arg]
```

$$\sum_{\ell=0}^{\ell\mathrm{Max}} \frac{b^{-1-\ell}\,q\,(b^{\ell} - r^{\ell})\,(1+2\,\ell)\,(1 + \mathrm{LegendreP}[\ell,-1])\,\mathrm{LegendreP}[\ell,\mathrm{Cos}[\theta]]}{2\,\ell\,(1+\ell)}$$

Note

```
Table[{ℓ, LegendreP[ℓ, -1], (-1)^ℓ}, {ℓ, 0, 5}] // ColumnForm
```

```
{0, 1, 1}
{1, -1, -1}
{2, 1, 1}
{3, -1, -1}
{4, 1, 1}
{5, -1, -1}
```

So

```
w4[10] = w4[9] /. LegendreP[ℓ, -1] → (-1)^ℓ
```

$$\sum_{\ell=0}^{\ell\text{Max}} \frac{\left(1+(-1)^{\ell}\right) b^{-1-\ell} q \left(b^{\ell}-r^{\ell}\right)(1+2\ell)\,\text{LegendreP}[\ell, \text{Cos}[\theta]]}{2\ell(1+\ell)}$$

There remains an issue — the first term (l = 0) is indeterminate.

Consider the summand.

```
w4[11] = w4[10][[1]]
```

$$\left(\left(1+(-1)^{\ell}\right) b^{-1-\ell} q \left(b^{\ell}-r^{\ell}\right)(1+2\ell)\,\text{LegendreP}[\ell, \text{Cos}[\theta]]\right) \Big/ \left(2\ell(1+\ell)\right)$$

Remove the terms that are well behaved as $l \to 0$

```
def["terms"] = terms == q (1 + (-1)^ℓ) LegendreP[ℓ, Cos[θ]] (1 + 2 ℓ) / (1 + ℓ)
```

$$\text{terms} == \frac{1}{1+\ell}\left(1+(-1)^{\ell}\right) q (1+2\ell)\,\text{LegendreP}[\ell, \text{Cos}[\theta]]$$

Check

```
def["terms"] /. ℓ → 0
```

$$\text{terms} == 2\,q$$

The problem terms are

```
w4[12] = w4[11] / terms /. (def["terms"] // ER)
```

$$\frac{b^{-1-\ell}\left(b^{\ell}-r^{\ell}\right)}{2\ell}$$

Determine the limiting value

```
w4[13] = Limit[w4[12], ℓ → 0]
```

$$\frac{\text{Log}[b]-\text{Log}[r]}{2\,b}$$

Then, the summand for l = 0 has value

```
w4[14] = (w4[13] terms) /. (def["terms"] // ER) /. ℓ → 0
```

$$\frac{q\,(\text{Log}[b]-\text{Log}[r])}{b}$$

So, to avoid difficulty with the summation

```
w4[15] = w4[14] + Sum[((1 + (-1)^ℓ) b^(-1-ℓ) q (b^ℓ - r^ℓ) (1 + 2 ℓ) LegendreP[ℓ, Cos[θ]])/(2 ℓ (1 + ℓ)), {ℓ, 1, ℓMax}]
```

$$\frac{q\,(\mathrm{Log}[b]-\mathrm{Log}[r])}{b}+\sum_{\ell=1}^{\ell\mathrm{Max}}\frac{\left(1+(-1)^{\ell}\right) b^{-1-\ell}\,q\,(b^{\ell}-r^{\ell})\,(1+2\,\ell)\,\mathrm{LegendreP}[\ell,\mathrm{Cos}[\theta]]}{2\,\ell\,(1+\ell)}$$

This should be compared with Jackson 3.136

I write a function for this expression.

```
Clear[ΦLineGreensFunction];
ΦLineGreensFunction[r_, θ_, q_, b_, ℓMax_] :=
 q (Log[b] - Log[r])/b + Sum[((1 + (-1)^ℓ) b^(-1-ℓ) q (b^ℓ - r^ℓ) (1 + 2 ℓ) LegendreP[ℓ, Cos[θ]])/(2 ℓ (1 + ℓ)), {ℓ, 1, ℓMax}]
```

For example

```
Activate@ΦLineGreensFunction[r, θ, q, b, 4]
```

$$\frac{1}{12\,b^{3}}5\,q\,(b^{2}-r^{2})\,\left(-1+3\,\mathrm{Cos}[\theta]^{2}\right)+\frac{1}{160\,b^{5}}9\,q\,(b^{4}-r^{4})\,\left(3-30\,\mathrm{Cos}[\theta]^{2}+35\,\mathrm{Cos}[\theta]^{4}\right)+\frac{q\,(\mathrm{Log}[b]-\mathrm{Log}[r])}{b}$$

For representative parameters and /Max = 30 terms in the expansion.

```
Module[{q = 1, b = 1, ℓMax = 30, conductor, lineCharge, range, lab, G},
 range = {{0, 1}, {0, 1}};
 lab = Stl["Φ from Green's function"];
 conductor = {Directive[Yellow, Thick], Circle[{0, 0}, b, {0, π/2}]};
 lineCharge = {Blue, Rectangle[{0, 0}, {0.03, b}], EdgeForm → Black };
 G[1] = Graphics[{conductor, lineCharge}, PlotRange → range, ImageSize → 250];
 G[2] = ContourPlot[
   Activate@ΦLineGreensFunction[Sqrt[x^2 + z^2], ArcTan[x / z], q, b, ℓMax], {x, 0, b},
   {z, 0, b}, FrameLabel → {{Rotate[Stl["z"], π/2], ""}, {Stl["x"], lab}},
   ColorFunction → "DeepSeaColors", Contours → 20, PlotRange → range,
   ImageSize → 250, RegionFunction → Function[{x, z}, x^2 + z^2 < b^2]];
 Show[G[2], G[1]]]
```



Appendix: Visualization

```
Module[{a = 1, b = 2, rhat = 1.5, CInner,
  COuter, CMatch, RInner, ROuter, RMatch, RLesser, RGreater},
 CInner = {Thick, Yellow, Circle[{0, 0}, a], {Black, Text["a", 1.1 {0, a}]}};
 COuter = {Thick, Yellow, Circle[{0, 0}, b], {Black, Text["b", 1.1 {b, 0}]}};
 CMatch = {Directive[Black, Dashed], Circle[{0, 0}, rhat],
   {Black, Text["r̂", 1.1 {rhat / √2, rhat / √2}]}};
 RInner = {Black, Arrow[{{0, 0}, {0, a}}]};
 ROuter = {Black, Arrow[{{0, 0}, {b, 0}}]};
 RMatch = {Black, Arrow[{{0, 0}, {rhat / √2, rhat / √2}}]};
 RLesser = With[{P = (a + rhat)/2 {Cos[π / 8], Sin[π / 8]}}, {Black, Arrow[{{0, 0}, P}],
    PointSize[0.02], Point[P], Text["r< case", 1.1 P + {-0.1, 0.15}]}];
 RGreater = With[{P = (b + rhat)/2 {Cos[3 π / 8], Sin[3 π / 8]}},
   {Black, Arrow[{{0, 0}, P}], PointSize[0.02],
    Point[P], Text["r> case", 1.1 P + {-0.1, 0.05}]}];
 Graphics[{CInner, COuter, CMatch, RInner, ROuter, RMatch, RLesser, RGreater},
  ImageSize → 300]]
```



```
Module[{a = 1, xMax = 2, zCharge = 1.75 ,
  rCharge = 0.05, vp = {2.4, 1.3, 2}, ptCharge, G},
 ptCharge = {Blue, Sphere[{0, 0, zCharge}, rCharge]};

 G[1] = Graphics3D[
   {White, Line[{{-xMax, 0, 0}, {xMax, 0, 0}}], Line[{{0, -xMax, 0}, {0, xMax, 0}}],
    Line[{{0, 0, -xMax}, {0, 0, xMax}}], Text["x", 1.1 {xMax, 0, 0}],
    Text["y", 1.1 {0, xMax, 0}], Text["z", 1.1 {0, 0, xMax}], ptCharge},
   Boxed → False, Background → Black, PlotLabel →
    Stl["Conducting sphere with external point charge"], ViewPoint → vp];
 G[2] = ContourPlot3D[{x^2 + y^2 + z^2 == a^2}, {x, -1, 1}, {y, -1, 1},
   {z, -1, 1}, Mesh → False, ViewPoint → Right,
   ContourStyle → Directive[Yellow, Opacity[0.8], Specularity[White, 30]],
   BoxRatios → Automatic, Boxed → False, Background → Black, ViewPoint → vp];
 Show[G[1], G[2]]]
```



```
Module[{a = 1, xMax = 2, rRing = 1.75,
  aRing = 0.05 , vp = {2.4, 1.3, 2}, ringCharge, range, G},
 range = {{-xMax, xMax}, {-xMax, xMax}, {-xMax, xMax}};
 G[0] = ParametricPlot3D[
   {Cos[ϕ] (rRing + aRing Cos[θ]), Sin[ϕ] (rRing + aRing Cos[θ]), aRing Sin[θ]},
   {ϕ, 0, 2 π}, {θ, 0, 2 π}, Mesh → False, PlotRange → range, PlotStyle → Blue,
   Background → Black, ViewPoint → vp, Boxed → False, Axes → False];

 G[1] = Graphics3D[{White, Line[{{-xMax, 0, 0}, {xMax, 0, 0}}],
    Line[{{0, -xMax, 0}, {0, xMax, 0}}], Line[{{0, 0, -xMax}, {0, 0, xMax}}],
    Text["x", 1.1 {xMax, 0, 0}], Text["y", 1.1 {0, xMax, 0}],
    Text["z", 1.1 {0, 0, xMax}]}, Boxed → False, Background → Black,
   PlotLabel → Stl["Conducting sphere with external point charge",
     PlotRange → range, ViewPoint → vp, Boxed → False]];
 G[2] = ContourPlot3D[{x^2 + y^2 + z^2 == a^2}, {x, -1, 1}, {y, -1, 1},
   {z, -1, 1}, Mesh → False, ContourStyle →
    Directive[Yellow, Opacity[0.8], Specularity[White, 30]], BoxRatios → Automatic,
   Boxed → False, Background → Black, PlotRange → range, ViewPoint → vp];
 Show[G[0], G[1], G[2]]]
```



```
Module[{a = 1, xMax = 2, rRing = 1.75,
  aRing = 0.05 , vp = {2.4, 1.3, 2}, ringCharge, range, G},
 range = {{-xMax, xMax}, {-xMax, xMax}, {-xMax, xMax}};

 G[1] = Graphics3D[
   {{White, Line[{{-xMax, 0, 0}, {xMax, 0, 0}}], Line[{{0, -xMax, 0}, {0, xMax, 0}}],
     Line[{{0, 0, -xMax}, {0, 0, xMax}}]}, {White, Text["x", 1.1 {xMax, 0, 0}],
     Text["y", 1.1 {0, xMax, 0}], Text["z", 1.1 {0, 0, xMax}]},
    {Directive[Blue, Thick], Line[{{0, 0, -a}, {0, 0, a}}] }},
   Boxed → False, Background → Black,
   PlotLabel → Stl["Conducting sphere with internal line charge"],
   PlotRange → range, ViewPoint → vp, Boxed → False];
 G[2] = ContourPlot3D[{x^2 + y^2 + z^2 == a^2}, {x, -1, 1}, {y, -1, 1}, {z, -1, 1}, Mesh →
    False, ContourStyle → Directive[Yellow, Opacity[0.8], Specularity[White, 30]],
   BoxRatios → Automatic, Boxed → False, Background → Black, PlotRange → range,
   ViewPoint → vp, RegionFunction → Function[{x, y, z}, x < 0 || y < 0]];
 Show[G[1], G[2]]]
```